WIS-CS-72-164
COMPUTER SCIENCES DEPARTMENT
The University of Wisconsin
1210 West Dayton Street
Madison, Wisconsin 53706

DISCRETE BARS, CONDUCTIVE HEAT TRANSFER, AND ELASTICITY

by

Donald Greenspan

Appendix: FORTRAN Program for Discrete Conductive Heat Transfer, by Sandie Turner Jones

Technical Report #164

November 1972

Received October 24, 1972

DISCRETE BARS, CONDUCTIVE HEAT TRANSFER, AND ELASTICITY

1. Introduction

Though the study of bars, or rods, is basic in structural analysis, heat transfer theory, and elasticity theory (see, e.g., refs. [1,3,4, 9-14] and the numerous references contained therein), most of the related models have been continuous and/or linear in nature. The intent of the present paper is to initiate a general computer oriented model which is discrete and nonlinear. For simplicity only, we will restrict attention to two dimensions, and, for convenience, we will describe an arithmetic, energy conserving n-body interaction model first.

2. Discrete n-Body Interaction

For positive time step Δt, let $t_k = k\Delta t$, $k = 0,1,2,\ldots$. At time t_k let particle P_i of mass m_i be located at $\vec{x}_{i,k} = (x_{i,k}, y_{i,k})$, have velocity $\vec{v}_{i,k} = (v_{i,k,x}, v_{i,k,y})$, and have acceleration $\vec{a}_{i,k} = (a_{i,k,x}, a_{i,k,y})$, for $i = 1,2,\ldots n$. Position, velocity, and acceleration are assumed to be related by the typical, discrete formulas [5,6]:

$$\frac{\vec{v}_{i,k+1} + \vec{v}_{i,k}}{2} = \frac{\vec{x}_{i,k+1} - \vec{x}_{i,k}}{\Delta t} \tag{2.1}$$

$$\vec{a}_{i,k} = \frac{\vec{v}_{i,k+1} - \vec{v}_{i,k}}{\Delta t} . \tag{2.2}$$

If $\vec{F}_{i,k} = (F_{i,k,x}, F_{i,k,y})$ is the force acting on P_i at time t_k, then force and acceleration are assumed to be related by the discrete dynamical equation

$$(2.3) \qquad \vec{F}_{i,k} = m_i \vec{a}_{i,k} \, .$$

In particular, we will choose $\vec{F}_{i,k}$ to have a component of attraction which behaves like $\frac{p}{r^\alpha}$ and a component of repulsion which behaves like $\frac{q}{r^\beta}$, where p, q, α and β are non-negative parameters with $\alpha \geq 2$, $\beta \geq 2$, and where r is the distance between a given pair of particles. For this purpose, let $r_{ij,k}$ be the distance between P_i and P_j at t_k. Then $\vec{F}_{i,k}$, the force exerted on P_i by the remaining particles, is defined by

$$(2.4) \qquad \vec{F}_{i,k} = m_i \sum_{\substack{j=1 \\ j \neq i}}^{n} \left\{ m_j \left(- \frac{p \sum_{\xi=0}^{\alpha-2} (r_{ij,k}^{\xi} \, r_{ij,k+1}^{\alpha-\xi-2})}{r_{ij,k}^{\alpha-1} \, r_{ij,k+1}^{\alpha-1} \, (r_{ij,k+1} + r_{ij,k})} + \frac{q \sum_{\xi=0}^{\beta-2} (r_{ij,k}^{\xi} \, r_{ij,k+1}^{\beta-\xi-2})}{r_{ij,k}^{\beta-1} \, r_{ij,k+1}^{\beta-1} (r_{ij,k+1} + r_{ij,k})} \right) (\vec{x}_{i,k+1} + \vec{x}_{i,k} - \vec{x}_{j,k+1} - \vec{x}_{j,k}) \right\} .$$

The particular value of (2.4) lies in the observation that if one defines system work W from t_0 to t_N by

$$W = \sum_{i=1}^{n} \sum_{k=0}^{N} [(\vec{x}_{i,k+1} - \vec{x}_{i,k}) \cdot \vec{F}_{i,k}],$$

system kinetic energy K_k at time t_k by

$$K_k = \sum_{i=0}^{n} [\frac{1}{2} m_i (v_{i,k,x}^2 + v_{i,k,y}^2)],$$

and system potential energy V_k at time t_k by

$$V_k = \sum_{\substack{i,j=1 \\ i<j}}^{n} \left[\left(- \frac{p}{r_{ij,k}^{\alpha-1}} + \frac{q}{r_{ij,k}^{\beta-1}} \right) m_i m_j \right],$$

then, as in [5] and [6],

$$K_N + V_N = K_0 + V_0, \quad N = 0,1,2,\ldots,$$

which is the classical law of conservation of energy.

3. The Solid State Building Block

In modeling a solid, we will attempt to simulate contemporary physical thought [2,8], in which molecules and atoms exhibit small vibrations within the solid. For this purpose, consider first a system of only two particles, P_1 and P_2, of equal mass, which interact according to (2.4). Assume that the force between the particles is zero. Then, from (2.4),

$$(3.1)\qquad \frac{-p\sum_{\xi=0}^{\alpha-2}\left(r_{ij,k}^{\xi}\, r_{ij,k+1}^{\alpha-\xi-2}\right)}{r_{ij,k}^{\alpha-1}\, r_{ij,k+1}^{\alpha-1}\,(r_{ij,k+1}+r_{ij,k})} + \frac{q\sum_{\xi=0}^{\beta-2}\left(r_{ij,k}^{\xi}\, r_{ij,k+1}^{\beta-\xi-2}\right)}{r_{ij,k}^{\beta-1}\, r_{ij,k+1}^{\beta-1}\,(r_{ij,k+1}+r_{ij,k})} = 0 .$$

But, if there is zero force between the two particles, then $r_{ij,k} = r_{ij,k+1}$, so set $r_{ij,k} = r_{ij,k+1} = r$ in (3.1) to yield

$$(3.2)\qquad \frac{-p\sum_{\xi=0}^{\alpha-2} r^{\alpha-2}}{r^{2\alpha-2}} + \frac{q\sum_{\xi=0}^{\beta-2} r^{\beta-2}}{r^{2\beta-2}} = 0 .$$

Thus, for $\beta \geq \alpha$,

$$- pr^{-\alpha}(\alpha-1) + qr^{-\beta}(\beta-1) = 0 ,$$

or, finally,

$$(3.3)\qquad r = \sqrt[\beta-\alpha]{\frac{q(\beta-1)}{p(\alpha-1)}} .$$

Consider next a system of only three particles, P_1, P_2 and P_3, of equal masses, whose mutual distances apart are given by (3.3). Since no force acts between any two of the particles, it follows that there is no force acting upon any one of the three. Such a configuration of particles is therefore exceptionally stable and will be called a triangular building block.

When considering a solid we will decompose it into triangular building blocks. In this fashion, the force on any particular particle due to its nearby neighbors will be zero. By an appropriate choice of parameters, the force on any particle due to more distant particles will be made small, thus achieving the small vibrations desired.

To illustrate, let the six particles $P_1, P_2, P_3, P_4, P_5, P_6$ be located at the vertices of the four triangular building blocks of the triangular region OAB, shown in Figure 3.1. Assume that $m_i \equiv 1$, $p = q = 1$, $\alpha = 7$, and $\beta = 10$, so that $r = \sqrt[3]{1.5}$. The particles' initial positions are, then,



Figure 3.1

P_1: (1.14471, 1.98270)

P_2: (0.57236, 0.99135)

P_3: (1.71707, 0.99135)

P_4: (0, 0)

P_5: (1.14471, 0)

P_6: (2.28943, 0) .

Assign to each particle a $\vec{0}$ initial velocity. Finally, let particles P_4 and P_6 be fixed and allow the remaining particles to move under force law (2.4). For $\Delta t = 0.05$ and for 2500 time steps, the motions of P_1, P_2, P_3 and P_5 were generated from (2.1)-(2.4). P_1 and P_5 exhibited small oscillations in the vertical directly only, while P_2 and P_3 exhibited small two dimensional oscillations. The maximum distance, for example, that P_1 moved from its initial position was approximately 0.02, and this occurred at approximately every one hundred time steps. The running time on the UNIVAC 1108 was 4 minutes. The basic computer program used, which is also typical of all examples which follow, was that of Jones [7].

Note that the magnitudes of the oscillations described above can be controlled completely by the appropriate choices of p, q, α, and β.

4. Flow of Heat in a Bar

Let us now develop the basic concepts of discrete conductive heat transfer by concentrating on the prototype problem of heat flow in a bar. Physically, the problem is formulated as follows. Let the region bounded by rectangle 0ABC, as shown in Figure 3.1, represent a bar. Let $|0A| = a$, $|0C| = c$. A section of the boundary of the bar is heated. The problem is to describe the flow of heat through the bar.

Our discrete approach to the problem proceeds as follows. First, subdivide the given region into triangular building blocks, one such possible subdivision of which is shown in Figure 3.2 for the parameter choices $m_i \equiv 1$, $p = q = 1$, $\alpha = 7$, $\beta = 10$, $a \sim 11$, $c \sim 2$. Note that from (3.3), $r \sim 1.1447142426$.

Now, by heating a section of the boundary of the bar, we will mean increasing the velocity, and hence the potential energy, of some of the particles whose centers are on 0ABC. By the temperature $T_{i,k}$ of particle P_i at time t_k, we will mean the following. Let M be a fixed positive integer and let $K_{i,k}$ be the kinetic energy of P_i at t_k. Then $T_{i,k}$ is defined by

$$T_{i,k} = \frac{1}{M} \sum_{j=k-M+1}^{k} K_{i,j},$$

which is, of course, the arithmetic mean of P_i's kinetic energies at M consecutive time steps. By the flow of heat through the bar we will mean the transfer to other particles of the bar of the kinetic energy added at the boundary. Finally, to follow the flow of heat through the bar one need only follow the motion of each particle and, at each time step, record its temperature.

To illustrate, consider the bar shown in Figure 3.2 with the parameter choices given above, that is, $m_i \equiv 1$, $p = q = 1$, $\alpha = 7$, $\beta = 10$, $a \sim 11$, $c \sim 2$. Assume that a strong heat source is placed above P_6, and then removed, in such a fashion that $\vec{v}_{5,0} = (\frac{-\sqrt{2}}{2}, \frac{-\sqrt{2}}{2})$, $\vec{v}_{6,0} = (0, -1)$, $\vec{v}_{7,0} = (\frac{\sqrt{2}}{2}, \frac{-\sqrt{2}}{2})$, while all other initial velocities are $\vec{0}$. With regard to temperature calculation, assume that the velocities of all particles prior to t_0 were $\vec{0}$. As regards the choice of M, which is a difficult choice to make, one would usually wish to choose it relatively large, since the use of an average is, generally, more meaningful when the number of quantities being averaged is relatively large. We shall arbitrarily set $M = 20$. From the resulting calculations with $\Delta t = 0.025$, Figures 3.3 - 3.7 show the constant temperature contours $T = 0.1, 0.06, 0.025, 0.002$ at t_5, t_{10}, t_{15}, t_{20} and t_{25}, respectively. The resulting wave motion is clear and Figure 3.7 exhibits wave reflection. It is interesting, also, to note that the temperature at P_6 increases, until t_{20}, at which time it is a maximum, and only then does it proceed to decrease. Figures 3.8 - 3.12 show the constant kinetic energy contours $K = 0.1, 0.05, 0.01, 0.001$ at each of the times t_5, t_{10}, t_{15}, t_{20}, t_{25}, respectively, and indicate the magnitude of the particle velocities at these time steps.

Other heat transfer concepts can be defined now in the same spirit as above, as follows. A side of the bar is insulated means that the bar particles cannot transfer energy across this side of the bar to particles outside the bar, while melting is the result of adding a sufficient quantity of heat so that various particle velocities attain sufficient magnitude so as to break the bonding effect of (2.4).



FIGURE 3.1



FIGURE 3.2



FIGURE 3.3



FIGURE 3.6



FIGURE 3.5



FIGURE 3.4



FIGURE 3.7



FIGURE 3.8



FIGURE 3.9



FIGURE 3.10



FIGURE 3.11



FIGURE 3.12

5. Oscillation of an Elastic Bar

Next, let us develop the basic concepts of discrete elasticity by concentrating on the vibration of an elastic bar. The problem is formulated physically as follows. Let the region bounded by rectangle 0ABC, as shown in Figure 3.1, represent a bar which can be deformed, and which, after deformation, tends to return to its original shape. The problem is to describe the motion of such a bar after the external force, which has deformed the bar, is removed. Equivalently, the problem is to describe the motion of an elastic bar after release from a position of tension.

Our discrete approach proceeds as follows. The given region is first subdivided into triangular building blocks. Then, deformation results in the compression of certain particles and the stretching apart of others. Release from a position of deformation, or tension, results, by (2.4), in repulsion between each pair of particles which have been compressed and attraction between each pair which have been stretched, the net effect being the motion of the bar.

As a particular example, let $m_i \equiv 1$, $\alpha = 7$, $\beta = 10$, $p = 425$, $q = 1000$, and $\Delta t = .025$. From (3.3), $r = 1.52254$. Consider, for variety, the thirty particle bar which results by deleting P_{11} and P_{32} from the configuration of Figure 3.2. The particles P_1, P_{12}, and P_{22}, whose respective coordinates are (0, 2.63711), (.76127, 1.31855), and (0,0), are to be held fixed throughout. In order to obtain an initial position of tension like that shown in Figure 4.1a, first set P_{13}, P_{14}, P_{15}, P_{16}, P_{17}, P_{18}, P_{19}, P_{20} and P_{21} at (2.28357, 1.29198), (3.80588, 1.26541), (5.32632, 1.18573), (6.84052, 1.02658), (8.33992, .76219), (9.81058, .36813), (11.23199, -.17750), (12.57631, -.89228), and (13.80807, -1.78721), respectively. Any two consecutive points P_k, P_{k+1}, $k = 13,14,\ldots,20$, are positioned r units apart. The points $P_2 - P_{10}$ and $P_{23} - P_{31}$ are then positioned as follows: P_{k-10} and P_{k+11} are the two points which are r units from both P_k and P_{k+1} for each of $k = 12,13,\ldots,20$. Each consecutive pair of points in the $P_2 - P_{10}$ set is then separated by a distance greater than r, while each consecutive pair of points in the $P_{23} - P_{31}$ set is separated by a distance less than r. Thus, the points $P_2 - P_{10}$ are in a stretched position, while the points $P_{23} - P_{31}$ are compressed.

From the initial position of tension shown in Figure 4.1a, the oscillatory motion of the bar is determined from (2.1)-(2.4) with all initial velocities set as $\vec{0}$. The upward swing of the bar was plotted automatically at every ten time steps and is shown in Figure 4.1a-w from t_0 to t_{220}. It is of interest to note that as the bar moves, each row of particles exhibits wave oscillation and reflection.

Fig. 4.1



(a) t_0



(b) t_{10}



(c) t_{20}



(d) t_{30}

Fig. 4.1 continued



(e) t_{40}

(f) t_{50}

(g) t_{60}

(h) t_{70}



(i) t_{80}

(j) t_{90}



(k) t_{100}



(l) t_{110}

Fig. 4.1 cont.



(m) t_{120}



(n) t_{130}



(o) t_{140}



(p) t_{150}

Fig. 4.1 cont.



(q) t_{160}



(r) t_{170}



(s) t_{180}



(t) t_{190}

Fig. 4.1 completed.



(u) t_{200}



(v) t_{210}



(w) t_{220}

6. Remarks

A limited number of other examples were run, and these indicated that square building blocks were less stable than triangular ones, while the choices $\alpha = 2$, $\beta = 5$ and $\alpha = 7$, $\beta = 13$ were less viable than $\alpha = 7$, $\beta = 10$. Generally speaking, any choice $p > q$ resulted in increased oscillations so that, for example, for the elastic bar model of Section 4, the choice $p = 3$, $q = 1$ required a refinement of time step to $\Delta t = 10^{-3}$ in order to study the resulting oscillations. The major handicap in all the computer examples run was the lack of adequate funding to enable the study of models with large numbers of particles.

Finally, it should be noted that the writer feels that varying α, β, p and q in computer models with large numbers of particles will enable the researcher to produce viable computer models and to derive insight into the actual parameter values for various physical solids.

References

1. J. Douglas, Jr., "A Survey of Numerical Methods for Parabolic Differential Equations," in Advances in Computers, vol. 2, Academic Press, N. Y., 1961, pp. 1-55.

2. R. P. Feynman, R. B. Leighton, and M. Sands, The Feynman Lectures on Physics, Addison-Wesley, Reading, Mass., 1964.

3. A. Friedman, Partial Differential Equations of Parabolic Type, Prentice-Hall, Englewood Cliffs, N. J., 1964.

4. D. Greenspan, Lectures on the Numerical Solution of Linear, Singular, and Nonlinear Differential Equations, Prentice-Hall, Englewood Cliffs, N. J., 1968.

5. ______________, "New Forms of Discrete Mechanics," Kybernetes, 1, 1972, pp. 87-101.

6. ______________, "An Algebraic, Energy Conserving Formulation of Classical Molecular and Newtonian n-Body Interaction," to appear in Bull. Amer. Math. Soc.

7. S. T. Jones, "FORTRAN program for discrete conductive heat transfer," Appendix, TR#164, Department of Computer Sciences, Univ. of Wis., Madison, 1972.

8. C. Kittel, Introduction to Solid State Physics, 4th Ed., Wiley, N. Y., 1971.

9. P. G. Klemens, "Theory of Thermal Conductivity of Solids," in Thermal Conductivity I, Academic Press, N. Y., 1969, pp. 1-68.

10. S. S. Kutateladze and V. M. Borishanskii, A Concise Encyclopedia of Heat Transfer, Pergamon, N. Y., 1966.

11. V. V. Novozhilov, Foundations of the Nonlinear Theory of Elasticity, Graylock, Rochester, N. Y., 1953.

12. W. Prager, Introduction to the Mechanics of Continua, Ginn, Boston, 1961.

13. S. Timoshenko and J. Gere, Theory of Elastic Stability, McGraw-Hill, N. Y., 1961.

14. S. Timoshenko and S. Woinowsky-Krieger, Theory of Plates and Shells, McGraw-Hill, N. Y., 1959.

APPENDIX: FORTRAN Program for Discrete Conductive Heat Transfer
by S. T. Jones

```
C   FORTRAN PROGRAM FOR DISCRETE CONDUCTIVE HEAT TRANSFER
C
C   INDEX TO PROGRAM VARIABLES
C
C      A = ALPHA
C      ANGLE(I) = ANGLE (IN DEGREES) OF INITIAL VELOCITY VECTOR WITH
C                 RESPECT TO POSITIVE X-AXIS
C      B = BETA
C      DD(I) = DISTANCE OF PARTICLE I FROM ITS INITIAL POSITION
C      DMAX = MAXIMUM ALLOWABLE DISTANCE OF ANY PARTICLE FROM ITS INITIAL
C             POSITION
C      DT = TIME INCREMENT
C      EPS = CONVERGENCE CRITERION FOR NEWTON'S METHOD
C      FX(I) = FORCE COMPONENT ON PARTICLE I IN X-DIRECTION
C      FY(I) = FORCE COMPONENT ON PARTICLE I IN Y-DIRECTION
C      IAXIS(I) = 0 IF LEFTMOST PARTICLE IN ROW I IS ON Y-AXIS
C               = +1,-1 IF LEFTMOST PARTICLE TO BE SHIFTED RIGHT OR LEFT,
C                         RESPECTIVELY
C      IEND = 0 IF ANOTHER DATA CASE FOLLOWS
C           = 1 IF END OF RUN
C      IMAX = MAXIMUM NUMBER OF ITERATIONS PER TIMESTEP FOR NEWTON'S
C             METHOD
C      IPRINT = PRINT-STEP INCREMENT
C      IPUNCH = PUNCH-STEP INCREMENT
C      IROW(I) = NUMBER OF PARTICLES IN ROW I
C      ISTART = 0 IF NEW DATA CASE
C             = 1 IF RESTART
C      IVEL(I) = NUMBER OF PARTICLE TO BE GIVEN AN INITIAL VELOCITY
C      JPUNCH = 0 IF NO PUNCH REQUIRED, PUNCH OTHERWISE
C      M(I) = MASS OF PARTICLE I
C      N = NUMBER OF PARTICLES IN SYSTEM
C      NAXIS = NUMBER OF ROW ON X-AXIS
C      NFIX = TOTAL NUMBER OF PARTICLES TO BE FIXED
C      NMAX = MAXIMUM NUMBER OF TIMESTEPS THIS DATA CASE
C      NP(I) = NUMBER OF PARTICLE TO BE FIXED
C      NROW = NUMBER OF ROWS IN SYSTEM
C      NSTP = TIMESTEP NUMBER
C      NVEL = TOTAL NUMBER OF PARTICLES TO BE GIVEN AN INITIAL VELOCITY
C      OMEGA = SUCCESSIVE OVER-RELAXATION FACTOR FOR NEWTON'S METHOD
C      P = ATTRACTION PARAMETER
C      Q = REPULSION PARAMETER
C      R(I,J,1) = DISTANCE BETWEEN PARTICLES I AND J AT PREVIOUS TIMESTEP
C      R(I,J,2) = DISTANCE BETWEEN PARTICLES I AND J AT CURRENT TIMESTEP
C      SKE(I) = SUM OF KINETIC ENERGIES FOR PARTICLE I OVER ALL TIMESTEPS
C      TEMP(I) = MEASURE OF TEMPERATURE OF PARTICLE I AT CURRENT TIMESTEP
C      VEL(I) = MAGNITUDE OF INITIAL VELOCITY VECTOR
C      VX(I,1) = X-COMPONENT OF VELOCITY OF PARTICLE I, PREVIOUS TIMESTEP
C      VX(I,2) = SAME AS ABOVE, CURRENT TIMESTEP, PREVIOUS ITERATION
C      VX(I,3) = SAME AS ABOVE, CURRENT TIMESTEP, CURRENT ITERATION
C      VX0(I) = X-COMPONENT OF INITIAL VELOCITY, PARTICLE I
C      VY(I,1) = Y-COMPONENT OF VELOCITY OF PARTICLE I,
C      VY(I,2) =       SAME DEFINITIONS
C      VY(I,3) =              AS VX(I,J), ABOVE
C      VY0(I) = Y-COMPONENT OF INITIAL VELOCITY, PARTICLE I
```

```
C     X(I,1) = X-COMPONENT OF POSITION OF PARTICLE I,
C     X(I,2) =          SAME DEFINITIONS
C     X(I,3) =              AS VX(I,J)
C     X0(I) = X-COMPONENT OF INITIAL POSITION, PARTICLE I
C     XKE(I) = KINETIC ENERGY OF PARTICLE I AT EACH TIMESTEP
C     Y(I,1) = Y-COMPONENT OF POSITION OF PARTICLE I,
C     Y(I,2) =          SAME DEFINITIONS
C     Y(I,3) =              AS VX(I,J)
C     Y0(I) = Y-COMPONENT OF INITIAL POSITION, PARTICLE I
C
      IMPLICIT DOUBLE PRECISION(A-H,M,O-Z)
      DIMENSION X0(100),Y0(100),VX0(100),VY0(100),X(100,3),Y(100,3),
     1VX(100,3),VY(100,3),FX(100),FY(100),DD(100),XKE(100),NP(100),
     2R(45,45,2),M(100)
      DIMENSION IROW(5),IAXIS(5),IVEL(5),VEL(5),ANGLE(5)
      DIMENSION SKE(50),TEMP(50)
 1001 FORMAT(8D10.0)
 1002 FORMAT(16I5)
 1003 FORMAT(4D10.0,I5)
 1004 FORMAT(I5,2D10.0)
 2000 FORMAT(1H1)
 2001 FORMAT(5X,'N',5X,'OMEGA',5X,'EPS',5X,'IMAX',/,I7,F10.4,E8.1,I9)
 2002 FORMAT(/,6X,'A',8X,'B',8X,'P',8X,'Q',7X,'DT',/,4F9.3,E9.2)
 2003 FORMAT(' TIMESTEP ',6X,'M',9X,'X',14X,'Y',13X,'DD',13X,'XKE',13X,
     1'SKE',13X,'TEMP',/,2X,I6)
 2004 FORMAT(10X,F6.4,4F15.10)
 2005 FORMAT('  NON-CONVERGENCE AFTER ',I3,' ITERATIONS FOR TIMESTEP= ',
     1I6,' DT= ',F9.2)
 2006 FORMAT(' OSCILLATION BEYOND MAXIMUM ALLOWABLE LIMIT OF ',F6.4)
 2501 FORMAT(2D25.18)
      PRINT 2000
      NSTP=0
      READ 1001,OMEGA,EPS,DMAX,DT
      READ 1002,NMAX,IMAX,IPRINT,IPUNCH,JPUNCH,ISTART
      READ 1002,N,NROW,NFIX,NAXIS,NVEL
      READ 1003,A,B,P,Q,TEND
      PRINT 2001,N,OMEGA,EPS,IMAX
      DO 20 I=1,N
      XKE(I)=0.0
      DD(I)=0.0
      SKE(I)=0.0
      TEMP(I)=0.0
   20 CONTINUE
      IF(ISTART.EQ.0)GO TO 1
C   RESTART
      READ 1002,NSTP
      READ 2501,(X0(I),Y0(I),I=1,N)
      READ 2501,(VX0(I),VY0(I),I=1,N)
      READ 2501,(SKE(I),I=1,N)
      GO TO 5
C   NEW CASE--CALCULATE POSITIONS
    1 BASEX=(Q*(B-1)/(P*(A-1)))**(1.0/(B-A))
      BASEY=SQRT(BASEX**2-(0.5*BASEX)**2)
      READ 1002,(IROW(I),IAXIS(I),I=1,NROW)
```

```
      IL=1
      IU=0
      DO 3  I=1,NROW
      IU=IU+IROW(I)
      XSHIFT=0.5*IAXIS(I)*BASEX
      JSHIFT=NAXIS-I
      DO 2  J=IL,IU
      X0(J)=(J-IL)*BASEX+XSHIFT
      Y0(J)=JSHIFT*BASEY
    2 CONTINUE
      IL=IU+1
    3 CONTINUE
C   NEW CASE--CALCULATE VELOCITIES
   10 DO 4  I=1,N
      VX0(I)=0.0
      VY0(I)=0.0
    4 CONTINUE
      IF(NVEL.EQ.0) GO TO 6
      READ 1004,(IVEL(I),VEL(I),ANGLE(I),I=1,NVEL)
      PI=3.14159265358979324D+00
      RAD=PI/180.0
      DO 5  I=1,NVEL
      J=IVEL(I)
      THETA=ANGLE(I)*RAD
      VX0(J)=VEL(I)*COS(THETA)
      VY0(J)=VEL(I)*SIN(THETA)
    5 CONTINUE
    6 DO 7  I=1,N
      M(I)=1.0
    7 CONTINUE
      OMW=1.0-OMEGA
      IF(NFIX.EQ.0)GO TO 11
      READ 1002,(NF(I),I=1,NFIX)
   11 PRINT 2002,A,B,H,W,DT
      T=0.0
      DT2=DT/2.0
      PRINT 2003,NSTP
      DO 30 I=1,N
      PRINT 2004,M(I),X0(I),Y0(I),D0(I),XKE(I)
   30 CONTINUE
C   SPECIFY INITIAL GUESS FOR NEWTON'S ITERATION AT FIRST TIMESTEP
      DO 40 I=1,N
      X(I,3)=X0(I)
      VX(I,3)=VX0(I)
      Y(I,3)=Y0(I)
   40 VY(I,3)=VY0(I)
      CALL RCALC
C   UPDATE POSITIONS,VELOCITIES,DISTANCES FOR ALL TIMESTEPS
   45 NSTP=NSTP+1
      T=T+DT
      DO 60 I=1,N
      X(I,1)=X(I,3)
      VX(I,1)=VX(I,3)
      Y(I,1)=Y(I,3)
```

```
      VY(I,1)=VY(I,3)
      DO 50 J=1,N
   50 R(I,J,1)=R(I,J,2)
   60 CONTINUE
C  BEGIN ITERATION LOOP
      DO 90 K=1,IMAX
C  UPDATE ALL VARIABLES, CURRENT TIMESTEP, PREVIOUS ITERATION
      DO 70 I=1,N
      X(I,2)=X(I,3)
      VX(I,2)=VX(I,3)
      Y(I,2)=Y(I,3)
      VY(I,2)=VY(I,3)
   70 CONTINUE
C  UPDATE POSITIONS, CURRENT TIMESTEP, CURRENT ITERATION
      DO 73 I=1,N
      IF(NFIX.EQ.0)GO TO 72
      DO 71 J=1,NFIX
      IF(I.EQ.NP(J))GO TO 73
   71 CONTINUE
   72 X(I,3)=OMW*X(I,2)+OMEGA*(DT2*(VX(I,2)+VX(I,1))+X(I,1))
      Y(I,3)=OMW*Y(I,2)+OMEGA*(DT2*(VY(I,2)+VY(I,1))+Y(I,1))
   73 CONTINUE
      CALL RCALC
      CALL FCALC
C  UPDATE VELOCITIES, CURRENT TIMESTEP, CURRENT ITERATION
      DO 80 I=1,N
      IF(NFIX.EQ.0)GO TO 75
      DO 74 J=1,NFIX
      IF(I.EQ.NP(J))GO TO 80
   74 CONTINUE
   75 VX(I,3)=OMW*VX(I,2)+OMEGA*(DT*FX(I)+VX(I,1))
      VY(I,3)=OMW*VY(I,2)+OMEGA*(DT*FY(I)+VY(I,1))
   80 CONTINUE
C  TEST FOR CONVERGENCE
      DO 85 I=1,N
      IF(ABS(X(I,3)-X(I,2)).GT.EPS)GO TO 90
      IF(ABS(Y(I,3)-Y(I,2)).GT.EPS)GO TO 90
      IF(ABS(VX(I,3)-VX(I,2)).GT.EPS)GO TO 90
      IF(ABS(VY(I,3)-VY(I,2)).GT.EPS)GO TO 90
   85 CONTINUE
      GO TO 95
   90 CONTINUE
      PRINT 2005,K,NSTEP
      GO TO 110
   95 CALL DCALC
      DO 100 I=1,N
      IF(DD(I).GT.DMAX)GO TO 105
  100 CONTINUE
      STEP=NSTEP
      DO 700 II=1,N
      XKE(II)=0.5*M(II)*(VX(II,3)*VX(II,3)+VY(II,3)*VY(II,3))
      SKE(II)=SKE(II)+XKE(II)
      TEMP(II)=SKE(II)/STEP
  700 CONTINUE
```

```
      IF(MOD(NSTP,IPRINT).NE.0)GO TO 103
      IF(JPUNCH.EQ.0)GO TO 102
      IF(MOD(NSTP,IPUNCH).NE.0)GO TO 102
      WRITE(1,1002)NSTP
      DO 151 I=1,N
      WRITE(1,2501)X(I,3),Y(I,3)
  151 CONTINUE
      DO 152 I=1,N
      WRITE(1,2501)VX(I,3),VY(I,3)
  152 CONTINUE
      WRITE(1,2501)(SKE(II),II=1,N)
  102 CALL OUTP
  103 IF(NSTP.EQ.NMAX)GO TO 110
      GO TO 95
  105 PRINT 2005,DMAX
      CALL OUTP
  110 IF(IEND.EQ.0)GO TO 10
      STOP
C  INTERNAL SUBROUTINE TO COMPUTE DISTANCES BETWEEN PARTICLES
      SUBROUTINE RCALC
      DO 210 II=1,N
      IP1=II+1
      DO 200 JJ=IP1,N
      R(II,JJ,2)=SQRT((X(II,3)-X(JJ,3))**2+(Y(II,3)-Y(JJ,3))**2)
      R(JJ,II,2)=R(II,JJ,2)
  200 CONTINUE
  210 CONTINUE
      RETURN
C  INTERNAL SUBROUTINE TO COMPUTE DISTANCE OF PARTICLE FROM ITS INITIAL
C  POSITION
      SUBROUTINE DCALC
      DO 300 II=1,N
      DD(II)=SQRT((X(II,3)-X0(II))**2+(Y(II,3)-Y0(II))**2)
  300 CONTINUE
      RETURN
C  INTERNAL SUBROUTINE TO COMPUTE FORCES
      SUBROUTINE FCALC
      IA=A-1
      IB=B-1
      DO 600 II=1,N
      IF(NFIX.EQ.0)GO TO 450
      DO 400 KK=1,NFIX
      IF(II.EQ.NP(KK))GO TO 600
  400 CONTINUE
  450 SUMX=0.0
      SUMY=0.0
  460 DO 550 JJ=1,N
      IF(II.EQ.JJ)GO TO 550
      SUMP=0.0
      SUMQ=0.0
      RTJ=R(II,JJ,1)+R(II,JJ,2)
      DO 500 IZ=1,IA
      SUMP=SUMP+(R(II,JJ,1)**(IZ-1))*(R(II,JJ,2)**(A-(IZ-1)-2))
  500 CONTINUE
```

```
      DO 501 I2=1,IB
      SUMQ=SUMQ+(R(II,JJ,1)**(I2-1))*(R(II,JJ,2)**(B-(I2-1)-2))
  501 CONTINUE
      PD=R(II,JJ,1)**(A-1)*R(II,JJ,2)**(A-1)*RIJ
      SUMP=P*SUMP/PD
      QD=R(II,JJ,1)**(B-1)*R(II,JJ,2)**(B-1)*RIJ
      SUMQ=Q*SUMQ/QD
      SUMX=(SUMQ-SUMP)*M(JJ)*(X(II,3)+X(II,1)-X(JJ,3)-X(JJ,1))+SUMX
      SUMY=(SUMQ-SUMP)*M(JJ)*(Y(II,3)+Y(II,1)-Y(JJ,3)-Y(JJ,1))+SUMY
  550 CONTINUE
      FX(II)=SUMX
      FY(II)=SUMY
  600 CONTINUE
      RETURN
C   INTERNAL PRINT SUBROUTINE
      SUBROUTINE OUTP
 3001 FORMAT(2X,I6)
 3002 FORMAT(15X,6F15.10)
      PRINT 3001,NSTP
      DO 800 II=1,N
      PRINT 3002,X(II,3),Y(II,3),DD(II),XKE(II),SKE(II),TEMP(II)
  800 CONTINUE
      RETURN
      END
```

BIBLIOGRAPHIC DATA SHEET	1. Report No. WIS-CS-72-164	2.	3. Recipient's Accession No.

4. Title and Subtitle Discrete Bars, Conductive Heat Transfer, and Elasticity	5. Report Date November 1972
	6.
7. Author(s) Donald Greenspan	8. Performing Organization Rept. No.
9. Performing Organization Name and Address Computer Sciences Department The University of Wisconsin 1210 West Dayton Street Madison, Wisconsin 53706	10. Project/Task/Work Unit No.
	11. Contract/Grant No.
12. Sponsoring Organization Name and Address	13. Type of Report & Period Covered
	14.

15. Supplementary Notes

16. Abstracts

Heat transfer and elasticity are modelled from a discrete point of view. Computer examples using bars are given to show the feasibility of the models. The dynamical equations are discrete and energy conserving.

17. Key Words and Document Analysis. 17a. Descriptors

Elasticity, heat transfer, numerical analysis.

17b. Identifiers/Open-Ended Terms

17c. COSATI Field/Group

18. Availability Statement Available to the public.	19. Security Class (This Report) UNCLASSIFIED	21. No. of Pages 30
	20. Security Class (This Page UNCLASSIFIED	22. Price